CAPÍTULO 8

( C. Castillo P. 



GRÁFICOS



	Un gráfico tiene por finalidad facilitar la legibilidad de la información que se presenta: “un gráfico vale más que mil palabras”.



LOS COLORES

En Visual Basic se pueden manipular los colores mediante tres formas diferentes: código hexadecimal, la función RGB() y la función QBColor().



Código Hexadecimal

	Un código de color se compone de seis dígitos hexadecimales, cuyos valores van desde &H000000& (cero) hasta &HFFFFFF& (16 777 215). De forma práctica podemos guiarnos mediante:



&HAAVVRR&



Donde AA representa al color azul, VV al color verde y RR al color rojo. Todos los demás colores se obtienen mezclando estos tres colores.

	Como ejemplo considere los siguientes códigos de colores que se presentan a continuación:



AZUL	 = &HFF0000&

VERDE = &H00FF00&

ROJO	 = &H0000FF&

NEGRO = &H000000&



La función RGB()

	La función RGB() retorna un número de tipo entero largo (Long) que representa el valor de un color. Su sintaxis es:



COLOR = RGB(nROJO, nVERDE, nAZUL)



Donde nROJO, nVERDE, nAZUL son enteros comprendidos entre 0 y 255. La siguiente tabla muestra algunos colores estándar y sus valores de nROJO, nVERDE y nAZUL:



�PRIVATE��Color�nROJO�nVERDE�nAZUL��Negro�0�0�0��Azul�0�0�255��Verde�0�255�0��Cián�0�255�255��Rojo�255�0�0��Magenta�255�0�255��Amarillo�255�255�0��Blanco�255�255�255��	El ejemplo anterior (código hexadecimal) utilizando la función RGB() será similar a:



AZUL	 = RGB(0, 0, 255)

VERDE = RGB(0, 255, 0)

ROJO	 = RGB(255, 0, 0)

NEGRO = RGB(0, 0, 0)



La función QBColor()

	La función QBColor() retorna un número de tipo entero largo (Long) que representa el código de color RGB correspondiente al número de color especificado. Su sintaxis es:



COLOR = QBColor(nCOLOR)



Donde el argumento nCOLOR es un entero entre 0 y 15; los valores que puede tomar se presentan en la siguiente tabla:



�PRIVATE��Número�Color�Número�Color��0�Negro�8�Gris��1�Azul�9�Azul claro��2�Verde�10�Verde claro��3�Cián�11�Cián claro��4�Rojo�12�Rojo claro��5�Magenta�13�Magenta claro��6�Amarillo�14�Amarillo claro��7�Blanco�15�Blanco brillante��	El ejemplo inicial (código hexadecimal) utilizando la función QBColor() se soluciona como se muestra a continuación:



AZUL	 = QBColor(1)

VERDE = QBColor(2)

ROJO	 = QBColor(4)

NEGRO = QBColor(0)



EL SISTEMA DE COORDENADAS

	Todas las operaciones gráficas descritas en este capítulo utilizan el sistema de coordenadas del área de dibujo (que puede ser la pantalla, un formulario, un cuadro de imagen, o cualquier otro contenedor).

El sistema de coordenadas es una cuadrícula bidimensional que define ubicaciones del área de dibujo. Las ubicaciones de esta cuadrícula se definen mediante las coordenadas del área de dibujo:



(� INCRUSTAR Equation.DSMT4  ���,� INCRUSTAR Equation.DSMT4  ���)



Donde el valor de � INCRUSTAR Equation.DSMT4  ��� es la ubicación del punto sobre el “eje X”, con la ubicación predeterminada (cero) en el extremo izquierdo. El valor de � INCRUSTAR Equation.DSMT4  ��� es la ubicación del punto sobre el eje � INCRUSTAR Equation.DSMT4  ���, con la ubicación predeterminada (cero) en el extremo superior. Este sistema de coordenadas se ilustra en la siguiente figura:

� INCRUSTAR PBrush  ���



El sistema de coordenadas de Visual Basic tiene las siguientes reglas: 



El sistema de coordenadas predeterminado de cualquier contenedor empieza por la coordenada (0, 0) en la esquina superior izquierda del contenedor. 

Las unidades de medida utilizadas para definir las ubicaciones a lo largo de los ejes se denominan colectivamente “escala”. En Visual Basic, cada eje del sistema de coordenadas puede tener su propia escala.

Todos los métodos gráficos utilizan el sistema de coordenadas del contenedor. Por ejemplo, las instrucciones que dibujan dentro de un cuadro de imagen utilizan el sistema de coordenadas de dicho control.



Establecer las coordenadas de dibujo

Para establecer las coordenadas de dibujo del contenedor se utilizan las propiedades CurrentX y CurrentY de dicho contenedor. 

Por ejemplo, las instrucciones siguientes establecen las coordenadas de dibujo a la esquina superior izquierda de un cuadro de imagen: 



Picture1.CurrentX = 0

Picture1.CurrentY = 0



De forma predeterminada, los formularios y los cuadros de imagen utilizan un sistema de coordenadas en el que cada unidad corresponde a un “twip” (1 440 twips equivalen a una pulgada y 567 twips son aproximadamente un centímetro). 



Cambiar la unidad de medida del sistema de coordenadas

Para cambiar la unidad de medida del sistema de coordenadas su establece la propiedad ScaleMode a uno de los valores enumerados en la tabla siguiente:



�PRIVATE��Valor de

ScaleMode�Descripción��1�Twips��2�Puntos��3�Pixeles��4�Caracteres��5�Pulgadas��6�Milímetros��7�Centímetros��Por ejemplo la siguiente orden establece la escala a pulgadas de un cuadro de imagen:



Picture1.ScaleMode = 5



Crear una escala personalizada

Para crear una escala personalizada se pueden utilizar las propiedades ScaleLeft, ScaleTop, ScaleWidth y ScaleHeight de un determinado objeto. 

Las propiedades ScaleLeft y ScaleTop asignan valores numéricos a la esquina superior izquierda de un objeto. Por ejemplo, las instrucciones siguientes establecen el valor de la esquina superior izquierda de un cuadro de imagen:



Picture1.ScaleLeft = 100

Picture1.ScaleTop = 100



Las propiedades ScaleWidth y ScaleHeight definen las unidades según el ancho y el alto real del área de dibujo (si el objeto cambia de tamaño posteriormente, las unidades siguen siendo las mismas). Por ejemplo las instrucciones siguientes definen la unidad horizontal como 1/1 000 del ancho interno actual del cuadro de imagen y la unidad vertical como 1/500 del alto interno actual del cuadro de imagen:



Picture1.ScaleWidth = 1000

Picture1.ScaleHeight = 500

A manera de ejemplo considere la escala mostrada en la siguiente figura, la cual tiene ScaleLeft, ScaleTop, ScaleWidth y ScaleHeight establecidas todas a 100.



� INCRUSTAR PBrush  ���



Una forma más eficiente de personalizar la escala de un sistema de coordenadas es utilizando el método Scale. Su sintaxis es:



OBJETO.Scale (X1, Y1) – (X2, Y2)



Donde los valores de (X1, Y1) determinan las coordenadas de la esquina superior izquierda, los valores de (X2, Y2) determinan los valores de la esquina inferior derecha. Para obtener el mismo sistema de coordenadas de ejemplo anterior (donde se utilizó las propiedades ScaleLeft, ScaleTop, ScaleWidth y ScaleHeight) la orden sería la siguiente:



Scale (100, 100)-(200, 200)

	Donde es evidente que las diferencias entre las dos coordenadas “X” y las dos coordenadas “Y” determinan los valores de ScaleWidth y ScaleHeight, respectivamente.



MÉTODOS GRÁFICOS

	Visual Basic proporciona varios métodos para la creación de gráficos. Los métodos gráficos, resumidos en la tabla siguiente, son aplicables a los formularios y a los cuadros de imagen.



�PRIVATE��Método�Descripción��Cls�Borra todos los gráficos y el resultado del método Print.��Pset�Dibuja un punto individual.��Point�Devuelve el color de un punto especificado.��Line�Dibuja una línea, un rectángulo o un cuadro relleno.��Circle�Dibuja una círculo, una elipse o un arco.��PaintPicture�Dibuja gráficos en ubicaciones arbitrarias.��Print�Imprime texto en formularios y cuadros de imagen.��

Para que la mayoría de los métodos gráficos funcionen correctamente la propiedad AutoRedraw del formulario debe está activada, caso contrario el dibujo no se visualizará.

Puntos

	Para el dibujo de puntos se usará el método Pset, cuya sintaxis es:



OBJETO.Pset (X, Y), COLOR



Donde X, Y son valores de tipo Single que indican las coordenadas horizontal (eje X) y vertical (eje Y) del punto que se va a dibujar. COLOR es un entero largo (Long) que indica el color RGB especificado para el punto, si se omite se usará el valor actual de la propiedad ForeColor de OBJETO.

Como ejemplo vamos a crear una sencilla aplicación que permita dibujar puntos en la posición especificada mediante el ratón, con la posibilidad de poder seleccionar colores (tipo “Paint”). 



� INCRUSTAR PBrush  ���

	Para el desarrollo de la presente aplicación proceda a crear un nuevo proyecto y agregar al formulario 34 cuadros de imagen. Luego ingresar el código mostrado.



Objeto�Nombre�Descripción��Picture1 . . . Picture32�PicColor�Opciones de color��Picture33�PicMuestra�Color actual utilizado��Picture34�PicLienzo�Area de dibujo��

Dim COLOR As Single

Private Sub Form_Load()

PicLienzo.ScaleMode = 3

PicLienzo.DrawWidth = 5

COLOR = RGB(0, 0, 0)

End Sub



Private Sub PicColor_Click(Index As Integer)

COLOR = PicColor(Index).BackColor

PicMuestra.BackColor = COLOR

End Sub



Private Sub PicLienzo_MouseDown(Button As Integer, _

Shift As Integer, X As Single, Y As Single)

If Button = vbLeftButton Then

PicLienzo.PSet (X, Y), COLOR

End If

End Sub

En el ejemplo anterior se utiliza la propiedad DrawWidth, la cual sirve para fijar el tamaño del punto dibujado.

Para que el método Pset funcione correctamente la propiedad ScaleMode debe estar establecida a pixeles.



Líneas, rectángulos y cuadros rellenos

	Para dibujar líneas, rectángulos o cuadros rellenos utilice el método Line, cuya sintaxis es:



OBJETO.Line (X1, Y1) - (X2, Y2), COLOR, B/BF



Donde X1, Y1 son valores de tipo Single que indican las coordenadas del punto inicial para la línea o el rectángulo. X2, Y2 son valores de tipo Single que indican las coordenadas del punto final para la línea o el rectángulo que se está dibujando. COLOR es un entero largo (Long) que indica el color RGB especificado para dibujar la línea, si se omite se usará el valor actual de la propiedad ForeColor de OBJETO. Los parámetros B y BF son opcionales: B hace que se dibuje un rectángulo usando las coordenadas para especificar las esquinas opuestas del mismo. BF especifica que el cuadro se rellena con el mismo color especificado mediante el parámetro COLOR.

	A manera de ejemplo vamos a mejorar nuestro sencillo “Editor gráfico”, de tal forma que además de dibujar puntos permita también dibujar líneas cuando se mantenga pulsado el botón izquierdo del ratón.

� INCRUSTAR PBrush  ���



Para lograr nuestro cometido añadir el siguiente código a la aplicación anterior:



Private Sub PicLienzo_MouseMove(Button As Integer, _

Shift As Integer, X As Single, Y As Single)

If Button = vbLeftButton Then

PicLienzo.Line -(X, Y), COLOR

End If

End Sub



	Otro ejemplo sencillo para ilustrar como se usa el método Line para dibujar rectángulos y cuadros rellenos se presenta a continuación, consiste en dibujar una bandera directamente sobre el formulario, en este caso será la bandera de mi país: Perú.

� INCRUSTAR PBrush  ���



Para este ejemplo crear una nueva aplicación e ingresar el código mostrado:



Private Sub Form_Click()

Height = 3 * 1440

Width = 5 * 1440

BackColor = QBColor(12)

ForeColor = QBColor(15)

Line (Width / 3, 0) - ((Width / 3) * 2, Height), , BF

End Sub



Círculos, elipses y arcos

	Para dibujar círculos, elipses o arcos se utiliza el método Circle, cuya sintaxis es:



OBJETO.Circle (X, Y), RADIO, COLOR, INICIO, FIN, ASPECTO

	Donde X, Y son valores de tipo Single que indican las coordenadas para el punto central del círculo, elipse o arco. RADIO es un valor de tipo Single que indica el radio del círculo, elipse o arco. COLOR es un valor entero de tipo Long que indica el color RGB del contorno del círculo, elipse o  arco (es opcional). INICIO y FIN son valores de tipo Single, que especifican (en radianes) las posiciones de inicio y de fin del arco (son opcionales). ASPECTO es un valor de tipo Single que indica la relación de aspecto del círculo, el valor predeterminado es 1.0, lo que produce un círculo perfecto (no elíptico) en cualquier pantalla (opcional).

	A manera de ejemplo vamos a mejor aún más nuestro “Editor gráfico” de tal manera que se podrá dibujar puntos, líneas, rectángulos, círculos y elipses.



� INCRUSTAR PBrush  ���



	Para el desarrollo de la presente aplicación se debe añadir al ejemplo anterior 5 botones de comandos y formar una matriz de controles (CmdForma). Luego ingresar el siguiente código:



Dim COLOR As Single

Dim FORMA As Integer

Dim X1 As Single, Y1 As Single, X2 As Single, Y2 As Single

Private Sub CmdForma_Click(Index As Integer)

FORMA = Index + 1

End Sub



Private Sub Form_Load()

COLOR = RGB(0, 0, 0)

FORMA = 1

PicLienzo.ScaleMode = 3

PicLienzo.DrawWidth = 3

End Sub



Private Sub PicColor_Click(Index As Integer)

COLOR = PicColor(Index).BackColor

PicMuestra.BackColor = COLOR

End Sub



Private Sub PicLienzo_MouseDown(Button As Integer, Shift As Integer, _

X As Single, Y As Single)

If Button = vbLeftButton Then

X1 = X

Y1 = Y

End If

End Sub

Private Sub PicLienzo_MouseUp(Button As Integer, Shift As Integer, _

X As Single, Y As Single)

If Button = vbLeftButton Then

X2 = X

Y2 = Y

RADIO = Sqr((X1 - X2) * (X1 - X2) + (Y1 - Y2) * (Y1 - Y2)) / 2

Select Case FORMA

Case 1

PicLienzo.PSet (X, Y), COLOR

Case 2

PicLienzo.Line -(X, Y), COLOR

Case 3

PicLienzo.Line (X1, Y1)-(X2, Y2), COLOR, B

Case 4

PicLienzo.Circle (X, Y), RADIO, COLOR

Case 5

PicLienzo.Circle (X, Y), RADIO, COLOR, , , 0.5

End Select

End If

End Sub



CONTROLES GRÁFICOS

	Visual Basic proporciona tres controles gráficos diseñados para crear efectos gráficos en las aplicaciones:



Control Image (estudiado en el Capítulo 4)

Control Line

Control Shape

Control Line

Line es un control gráfico que se muestra como una línea horizontal, vertical o diagonal.

El control Line tiene una funcionalidad limitada y está pensado para usos sencillos. Para usos más avanzados debe usar el método Line.



Propiedad�Descripción��BorderColor�Devuelve o establece el color de la línea.��BorderStyle�Devuelve o establece el estilo de la línea. ��BorderWidth�Devuelve o establece el ancho de la línea.��X1, Y1, X2, Y2�Devuelven o establecen las coordenadas del punto inicial (X1, Y1) y del punto final   (X2, Y2) de la línea.��

Control Shape

Shape es un control gráfico que se muestra como un rectángulo, un cuadrado, una elipse, un círculo, un rectángulo redondeado o un cuadrado redondeado.

El control Shape presenta propiedades similares al control Line (BorderColor, BorderStyle, BorderWidth). Adicionalmente, la propiedad Shape del control Shape proporciona seis formas predefinidas. En la tabla siguiente se enumeran todas las formas predefinidas, sus valores y las constantes de Visual Basic equivalentes:

�PRIVATE��Forma�Estilo�Constante��Rectángulo�0�vbShapeRectangle��Cuadrado�1�vbShapeSquare��Elipse�2�vbShapeOval��Círculo�3�vbShapeCircle��Rectángulo redondeado�4�vbShapeRoundedRectangle��Cuadrado redondeado�5�vbShapeRoundedSquare���	Las seis formas predefinidas que presenta el control Shape se muestran en el siguiente formulario:



� INCRUSTAR PBrush  ���



GRAFICACIÓN MATEMÁTICA

	Como la mayor parte de las ecuaciones matemáticas expresa relaciones complicadas en una, dos tres o más dimensiones, tratar de entenderlas sin gráficas es casi lo mismo que tratar de caminar con los ojos vendados.

Funciones algebraicas

	Como ejemplo de este tipo de gráficas vamos a dibujar una parábola, cuya ecuación matemática (ecuación general de segundo grado) es de la forma:



� INCRUSTAR Equation.DSMT4  ���



� INCRUSTAR PBrush  ���



Para realizar la presente aplicación crear un nuevo proyecto y añadir al formulario un cuadro de imagen (PicPlano), tres cuadros de texto (TxtA, TxtB, TxtC) y un botón de comandos (CmdVisualizar). Los aspectos decorativos (etiquetas, marcos, etc.) quedan a criterio del lector. Luego ingresar el siguiente código:

Dim A As Double, B As Double, C As Double

Function F(ByVal X As Single) As Single

F = A * (X * X) + B * X + C

End Function



Private Sub CmdVisualizar_Click()

Dim X As Single

A = TxtA

B = TxtB

C = TxtC

PicPlano.Cls

PicPlano.Scale (-100, 100)-(100, -100)

PicPlano.Line (-100, 0)-(100, 0)

PicPlano.Line (0, -100)-(0, 100)

PicPlano.CurrentX = -10

PicPlano.CurrentY = F(-10)

For X = -10 To 10

     	PicPlano.Line -(X, F(X))

Next

End Sub



Funciones trigonométricas

	Graficar funciones trigonométricas (seno, coseno, tangente, cotangente, secante, cosecante) no es nada complicado, sino analice el siguiente ejemplo.

� INCRUSTAR PBrush  ���



Para realizar nuestra aplicación de ejemplo crear un nuevo proyecto y añadir al formulario un cuadro de imagen (PicPlano) y 6 botones de comandos (CmdFuncion, es decir se trata de una matriz de controles). Luego ingresar el código siguiente:



Private Sub CmdFuncion_Click(Index As Integer)

    PicPlano.Cls

    Dibuja (Index)

End Sub



Private Sub Dibuja(OP As Integer)

PicPlano.Scale (-5, 5)-(5, -5)

PicPlano.Line (-5, 0)-(5, 0)

PicPlano.Line (0, -5)-(0, 5)

DosPi = 8 * Atn(1)

Dim X As Single

For X = -DosPi To DosPi Step 0.025

Select Case OP

Case 0

Y = Sin(X)

Case 1

Y = Cos(X)

Case 2

Y = Sin(X) / Cos(X)

Case 3 

If X < -0.001 Or X > 0.001 Then

                    	Y = Cos(X) / Sin(X)

End If

Case 4

Y = 1 / Cos(X)

Case 5

If X < -0.001 Or X > 0.001 Then

Y = 1 / Sin(X)

End If

End Select

PicPlano.PSet (X, Y)

Next

End Sub



Gráficas en coordenadas polares

	Para el caso de las gráficas de ecuaciones polares tampoco es complicado, sólo hay que tener presente las siguientes consideraciones.

	Dado un punto � INCRUSTAR Equation.DSMT4  ��� del plano, tal punto se encuentra a una distancia � INCRUSTAR Equation.DSMT4  ��� del origen, y a un ángulo � INCRUSTAR Equation.DSMT4  ��� respecto del semieje horizontal positivo.



� INCRUSTAR PBrush  ���



De la gráfica se tiene que el punto � INCRUSTAR Equation.DSMT4  ��� tiene coordenadas rectangulares � INCRUSTAR Equation.DSMT4  ��� y coordenadas polares � INCRUSTAR Equation.DSMT4  ��� entonces se cumplen las siguientes relaciones de transformación:



� INCRUSTAR Equation.DSMT4  ���		

� INCRUSTAR Equation.DSMT4  ���



De donde: � INCRUSTAR Equation.DSMT4  ��� ; � INCRUSTAR Equation.DSMT4  ���

A continuación presentamos algunas ecuaciones polares características:



Nombre�Ecuación��Circunferencia de radio � INCRUSTAR Equation.DSMT4  ����� INCRUSTAR Equation.DSMT4  �����Trébol de cuatro hojas�� INCRUSTAR Equation.DSMT4  �����Cardioide�� INCRUSTAR Equation.DSMT4  �����Limazón�� INCRUSTAR Equation.DSMT4  �����Lemniscata�� INCRUSTAR Equation.DSMT4  �����Pétalo de rosa�� INCRUSTAR Equation.DSMT4  �����Espiral�� INCRUSTAR Equation.DSMT4  �����Espiral logarítmica�� INCRUSTAR Equation.DSMT4  �����Astroide�� INCRUSTAR Equation.DSMT4  �����Nefroide�� INCRUSTAR Equation.DSMT4  �����Concoide�� INCRUSTAR Equation.DSMT4  �����Estrofoide�� INCRUSTAR Equation.DSMT4  �����

	Para realizar una aplicación que permita dibujar ecuaciones polares crear un nuevo proyecto y añadir al formulario 1 cuadro de imagen (PicPlano), un cuadro de lista (LstEcuacion) y un botón de comandos (CmdVisualizar). La apariencia de la interfaz debe ser similar a la de la figura mostrada:

� INCRUSTAR PBrush  ���



Luego proceda a ingresar el código que se presenta a continuación:



Dim OP As Integer

Function F(ByVal THETA As Single) As Single

Select Case OP

Case 0 : F = 2

Case 1 : F = 2 * Cos(2 * THETA)

Case 2 : F = 1 + Cos(THETA)

Case 3 : F = 1 + 2 * Cos(THETA)

Case 4 : If Cos(2 * THETA) >= 0 Then F = Sqr(Cos(2 * THETA))

Case 5 : F = Sin(4 * THETA)

Case 6 : F = THETA / 2

Case 7 : F = Exp(THETA / 8)

Case 8 : F = Cos(2 * THETA) * Cos(2 * THETA) * Cos(2 * THETA)

Case 9 : F = 3 * Cos(2 * THETA) - Cos(6 * THETA)

Case 10: F = 1 / Cos(2 * THETA) + 1

Case 11: If Int(THETA / 45) <> THETA / 45 Or THETA <> 0 _

Then F = 1 / (Cos(2 * THETA) + Tan(2 * THETA))

End Select

End Function



Private Sub LstEcuacion_Click()

OP = LstEcuacion.ListIndex

End Sub



Private Sub CmdVisualizar_Click()

Dim X As Single, Y As Single

Dim I As Single, R As Single

DosPi = 8 * Atn(1)

PicPlano.Cls

PicPlano.ScaleMode = 3

PicPlano.Scale (-3.25, 3.25)-(3.25, -3.25)

PicPlano.Line (-3.25, 0)-(3.25, 0)

PicPlano.Line (0, -3.25)-(0, 3.25)

For I = 0 To DosPi Step 0.01

R = F(I)

X = R * Cos(I)

Y = R * Sin(I)

PicPlano.PSet (X, Y), QBColor(1)

Next

End Sub

Fractales

La matemática fractal es producto de la interacción de un matemático genial (Benoit B. Mandelbrot) y la computadora.

El ahora ubicuo término fractal se debe al hecho de que la matemática fractal trabaja con fracciones, las figuras generadas no tienen una relación de enteros entre sus componentes, de modo que ninguna dimensión es producto de otra multiplicada por un entero. Es decir, al dividir una dimensión entre otra nunca se obtiene un resultado exacto. Los fractales generan figuras y estas figuras resultan tener una extraña similitud con algunas formas que genera la naturaleza. 

	Entre las ecuaciones para generar fractales Mandelbrot ha producido una en particular que ha recibido el nombre de “Conjunto de Mandelbrot”. Esta ecuación se resuelve en el campo de los números comprendidos entre las longitudes -2.0 a +0.5 y las latitudes -1.25 a +1.25, está dominado por un gran mar fractal con bahías y entrantes. La expresión matemática del Conjunto de Mandelbrot es: 



� INCRUSTAR Equation.DSMT4  ���



Donde � INCRUSTAR Equation.DSMT4  ��� es un número complejo constante y � INCRUSTAR Equation.DSMT4  ��� es un número complejo variable que permite obtener las respectivas iteraciones (los valores de cada punto del paisaje fractal, latitud y longitud).

	Como ejemplo vamos a obtener las tres primeras iteraciones. Para ello asignamos valores a � INCRUSTAR Equation.DSMT4  ��� de cualquier posición arbitraria dentro del conjunto, digamos � INCRUSTAR Equation.DSMT4  ��� y para arrancar las iteraciones podemos comenzar con � INCRUSTAR Equation.DSMT4  ���, luego:



� INCRUSTAR Equation.DSMT4  ���

� INCRUSTAR Equation.DSMT4  ���

� INCRUSTAR Equation.DSMT4  ���



	Cuando se continua desarrollando el Conjunto de Mandelbrot se obtiene una figura similar a la mostrada:



�



	Para el desarrollo de la presente aplicación crear un nuevo proyecto y añadir sobre el formulario un cuadro de imagen (PicPlano) y un botón de comandos (CmdVisualizar), luego ingresar el código siguiente:



Function Hipotenusa(ByVal A As Double, ByVal B As Double) As Double

Hipotenusa = Sqr(A * A + B * B)

End Function



Private Sub CmdVisualizar_Click()

Dim MaxX As Integer, MaxY As Integer

Dim Limite As Integer, Pasos As Integer

Dim I As Integer, J As Integer

Dim PasoX As Double, PasoY As Double

Dim PosX As Double, PosY As Double

Dim OrigX As Double, OrigY As Double

Dim DimX As Double, DimY As Double

Dim IterX As Double, IterY As Double, TempX As Double

Dim Terminar As Boolean

PicPlano.ScaleMode = 3

MaxX = PicPlano.ScaleWidth

MaxY = PicPlano.ScaleHeight

Limite = 16

OrigX = -2

OrigY = -1.25

DimX = 0.5

DimY = 1.25

PasoX = (DimX - OrigX) / MaxX

PasoY = (DimY - OrigY) / MaxY

For I = 0 To MaxX

For J = 0 To MaxY

PosX = OrigX + I * PasoX

PosY = OrigY + J * PasoY

IterX = 0  :  IterY = 0

Terminar = False

Pasos = 0

While (Not Terminar)

    TempX = (IterX * IterX) - (IterY * IterY) + PosX

                    IterY = 2 * (IterX * IterY) + PosY

                    IterX = TempX

                    Pasos = Pasos + 1

                    If Hipotenusa(Abs(IterX), Abs(IterY)) >= 2 Then

Terminar = True

    End If

    If (Pasos >= Limite) Then

Terminar = True

    End If

Wend

If (Pasos < Limite) Then

    PicPlano.PSet (I, J), QBColor(Pasos)

End If

Next

Next

End Sub

	En este capítulo, debido a las limitaciones de espacio y materia, tan sólo se ofrece el camino de acceso a unos pocos de los diversos temas que abarca el fascinante mundo de la programación de gráficos por computadora
.
 



	Muchos de estos temas siguen siendo materia de investigación (como el caso de los fractales), por lo que si Ud. desea profundizar en este laberinto de fascinación, no hay la garantía de un retorno sano y salvo a nuestro universo mundano.

